



	******************************************************************
	**
	**
	**		NAME:	ELITE PERSISTENCE IN THE ERA OF EXPANDING OVERSEAS TRADE
	**
	**		MAKES:  Figures in Appendix A		       
	**		
	**
	******************************************************************
	
	
	
	
	

			

		
			**	**	**	**	**	**	**	**	**	**	**	**	**
			**	**	**	**	**	**	**	**	**	**	**	**	**
			**	**	**	**	**	**	**	**	**	**	**	**	**
	


*-------------------------------------------------------------------------------
* global for input/output
*-------------------------------------------------------------------------------


* input
*------

global input = "" 	
	
	
* output
*-------

global output = ""	
		
	
	
			**	**	**	**	**	**	**	**	**	**	**	**	**
			**	**	**	**	**	**	**	**	**	**	**	**	**
			**	**	**	**	**	**	**	**	**	**	**	**	**
	


*-------------------------------------------------------------------------------
* preliminaries
*-------------------------------------------------------------------------------


* preliminaries
*--------------

clear
set more off







	
			**	**	**	**	**	**	**	**	**	**	**	**	**
			**	**	**	**	**	**	**	**	**	**	**	**	**
			**	**	**	**	**	**	**	**	**	**	**	**	**
	


*-------------------------------------------------------------------------------
* plot: commercial output (FIGURE A1)
*-------------------------------------------------------------------------------


* use
*----

cd ${input}
use analysis_vars_othertrends.dta, clear



* scatter
*--------

gen tt_scatter = trade_transport_perhead ///
			if parl_year == 1275 | parl_year == 1305 | parl_year == 1355 ///
			| parl_year == 1405 | parl_year == 1455 | parl_year == 1505 ///
			| parl_year == 1555 | parl_year == 1605 | parl_year == 1655 ///
			| parl_year == 1705 | parl_year == 1750
			
			

* hi lo
*------

capture drop hi lo
gen hi = 2000
gen lo = 0
sort parl_year




* plot
*-----

# delimit ;

twoway 
		
	(rarea hi lo parl_year	
		if   parl_year >= 1550 & parl_year <= 1750 
		,  color(gs12) )
		
	(line trade_transport_perhead  parl_year
		if   parl_year <= 1750	& parl_year >= 1250	
		,  lcolor(gs1)  lwidth(medthick) lpattern(line)    )	
	
	(scatter tt_scatter  parl_year
		if   parl_year <= 1750		& parl_year >= 1250		
		,  mcolor(gs1)  msize(medsmall)     )
	
		,
		yscale(noline )
		xscale(noline )
		
		xsize(8.5)
		ysize(3.0)
		

		title("", color(black) size(medsmall) pos(11) ) 
		xlab( 1250(100) 1750, 
			labsize(medsmall) ) 
		ylab(, 
			angle(hori) labsize(medsmall) nogrid  )
		xtitle("Year" , 
			color(black) size(large) )
		ytitle("Trade & Transport Output (1300=100)", 
			color(black) size(large)  )
		graphregion( fcolor(white) lcolor(white) ) 
		plotregion( fcolor(white) lstyle(none) lcolor(white) ilstyle(none)) 
		legend (order(
				2 "Trade & Transport Output"
				)
			cols(1)
			pos(11)
			ring(0)
			region( color(none) )
			size(large)
			)
		
		;

# delimit cr


	* export graph
	*-------------

	cd "${output}"
	graph export "Trend_Trade_Output.pdf", replace
	
	
	
	
	
	

			**	**	**	**	**	**	**	**	**	**	**	**	**
			**	**	**	**	**	**	**	**	**	**	**	**	**
			**	**	**	**	**	**	**	**	**	**	**	**	**
	


*-------------------------------------------------------------------------------
* PLOT: domestic and intl trade (FIGURE A2)
*-------------------------------------------------------------------------------



* use
*----

cd ${input}
use analysis_vars_othertrends.dta, clear



* hi lo
*------

capture drop hi lo
gen hi = 100
gen lo = 0
sort parl_year


 
* plot
*-----

# delimit ;

twoway 
		
	(rarea hi lo parl_year	
		if   parl_year >= 1550 & parl_year <= 1700 
		,  color(gs12) )
		
	(line trade_intl  parl_year
		if   parl_year <= 1700		
		,  lcolor(gs1)  lwidth(medthick) lpattern(line)    )	
	
		
	(line trade_domestic  parl_year
		if   parl_year <= 1700		
		,  lcolor(gs5)  lwidth(medthick) lpattern(shortdash)    )	
	

	
		,
		yscale(noline )
		xscale(noline )
		
		xsize(8.5)
		ysize(3.0)
		

		title("", color(black) size(medsmall) pos(11) ) 
		xlab( 1250(100) 1750, 
			labsize(medsmall) ) 
		ylab(, 
			angle(hori) labsize(medsmall) nogrid  )
		xtitle("Year" , 
			color(black) size(large) )
		ytitle("Trade Output (1700=100)", 
			color(black) size(large)  )
		graphregion( fcolor(white) lcolor(white) ) 
		plotregion( fcolor(white) lstyle(none) lcolor(white) ilstyle(none)) 
		legend (order(
				2 "International Trade"
				3 "Domestic Trade"
				)
			cols(1)
			pos(11)
			ring(0)
			region( color(none) )
			size(large)
			)
		
		;

# delimit cr


	* export graph
	*-------------

	cd "${output}"
	graph export "Trend_TradeType_Output.pdf", replace
	

	
	


			**	**	**	**	**	**	**	**	**	**	**	**	**
			**	**	**	**	**	**	**	**	**	**	**	**	**
			**	**	**	**	**	**	**	**	**	**	**	**	**
	


*-------------------------------------------------------------------------------
* PLOT: bairoch cities (FIGURE A3)
*-------------------------------------------------------------------------------


	
* use
*----

cd ${input}
use analysis_vars_bairoch.dta, clear
	
	
	

* hi lo
*------

capture drop hi lo
gen hi = 6
gen lo = 0




* sort
*-----

sort year



* plot
*-----

# delimit ;

twoway 
	
	(rarea hi lo year	
		if   year >= 1500 & year <= 1750 
		,  color(gs14) )
		
		
		// atlantic not london
	(line pop_atl_nl_avg year
		if tag_atlantic == 1 & id_city == 161	& year < 1800	
		,  lcolor(gs1)  lwidth(medium) lpattern(line)   )
	
	(scatter pop_atl_nl_avg year
		if   tag_atlantic == 1 	& id_city == 161	& year < 1800	
		,  mcolor(gs1) msize(medsmall) msymbol(circle))
		
	
		
		
		// not atlantic
	(line pop_natl_avg year
		if tag_atlantic != 1 	& id_city == 163	& year < 1800		
		,  lcolor(gs5)  lwidth(medium) lpattern(shortdash)  )
	(scatter pop_natl_avg year
		if   tag_atlantic != 1 	& id_city == 163	& year < 1800	
		,  mcolor(gs5) msize(medsmall) msymbol(dh) )		
	


	,
		yscale(noline )

		xscale(noline )

		title("", color(black) size(medsmall) pos(11) ) 
		xlab( , 
			labsize(medsmall) ) 
		ylab( , 
			angle(hori) labsize(medsmall) nogrid  )

		xtitle("Year" , 
			color(black) size(medium) )
		ytitle("Avg. Population (,000)", 
			color(black) size(medium)  )

		graphregion( fcolor(white) lcolor(white) ) 
		plotregion( fcolor(white) lstyle(none) lcolor(white) ilstyle(none)) 
		legend (order(
				
				2 "Cities <5km Ocean (excl. London)"
				4 "Cities >5km Ocean"
				)
			cols(1)
			pos(11)
			ring(0)
			region( color(none) )
			size(medium)
			)
		
		;

# delimit cr


	* export graph
	*-------------

	cd "${output}"
	graph export "Trend_Cities.pdf", replace
	
	
		

			**	**	**	**	**	**	**	**	**	**	**	**	**
			**	**	**	**	**	**	**	**	**	**	**	**	**
			**	**	**	**	**	**	**	**	**	**	**	**	**
	


*-------------------------------------------------------------------------------
* PLOT: eic voyages (FIGURE A4)
*-------------------------------------------------------------------------------


* use
*----

cd ${input}
use analysis_vars_othertrends.dta, clear



* sort
*-----

capture drop hi lo
gen hi = 40
gen lo = 0



* plot
*-----

# delimit ;

twoway 
		// rarea
	(rarea hi lo parl_year	
		if   parl_year >= 1640 & parl_year <= 1660 
		,  color(gs12) )
		
	
	(line eic_avg_annual_voy parl_year
		if parl_year >= 1600 & parl_year <= 1750		
		,  lcolor(gs2)  lwidth(medthick) lpattern(line)   )
	
	(line eic_avg_annual_tonnage parl_year
		if  parl_year >= 1600 & parl_year <= 1750		
		,  lcolor(gs7)  lwidth(medthick) lpattern(dash)  yaxis(2) )
		
		,
		yscale(noline  axis(1) )
		yscale(noline  axis(2) )
		xscale(noline  )
		
		xsize(8.5)
		ysize(3.0)
		

		title("", color(black) size(medsmall) pos(11) ) 
		xlab( 1600(50)1750, 
			labsize(medsmall) ) 
		ylab(, 
			angle(hori) labsize(medsmall) nogrid axis(1) )
		ylab(, 
			angle(hori) labsize(medsmall) nogrid axis(2) )
			
		xtitle("Year" , 
			color(black) size(large) )
		ytitle("Avg. annual voyages", 
			color(black) size(large) axis(1) )
		ytitle("Avg. annual tonnage voyaging", 
			color(black) size(large) axis(2) )
		graphregion( fcolor(white) lcolor(white) ) 
		plotregion( fcolor(white) lstyle(none) lcolor(white) ilstyle(none)) 
		legend (order(
				2 "Avg. annual voyages"
				3 "Avg. annual tonnage voyaging"
				)
			cols(1)
			pos(11)
			ring(0)
			region( color(none) )
			size(medium)
			symx(*.68)
			)
		
		;

# delimit cr



* export graph
	*-------------

	cd "${output}"
	graph export "Trend_EIC_Voyages.pdf", replace
	
	
	
	
	






			**	**	**	**	**	**	**	**	**	**	**	**	**
			**	**	**	**	**	**	**	**	**	**	**	**	**
			**	**	**	**	**	**	**	**	**	**	**	**	**
	

						** end of do file ** 	
	
	
	